LINEAR PROGRAMMING MODELING EXAMPLES:
Multiperiod Planning

multi period production planning

The National Steel Corporation (NSC) produces a special-purpose steel used in aerospace industries. The sales department has received orders for the next four months:

	Jan	Feb	Mar	Apr
Demand (tons)	2400	2200	2700	2500

NSC can meet these demands by producing the steel, by drawing from its inventory or by a combination of both. January inventory is 1000.

multi period production planning

The steel production costs per ton vary from month to month – projections are:

	Jan	Feb	Mar	Apr
Production cost	**7400**	**7500**	**7600**	**7800**
Inventory cost	**120**	**120**	**120**	**120**

Monthly production capacity is 4000 tons.

Operations requires ending inventory for April to be 1500 tons.

A production plan, i.e., the amount of steel to produce in each of the next 4 months.

Minimize the total production and inventory cost.
The costs must be calculated from the decision variables.

What needs to be decided?

What is the objective?

What are the constraints?

Demand must be met each month.
Constraints to define inventory in each month.
Production-capacity constraints.
Non-negativity of the production and inventory quantities.

Decision Variables

- Let P_i be the tons of steel produced in month i
- Let I_i be the tons of steel in inventory at the end of month i.
  - Note: The initial inventory is $I_0 = 1000$

Objective Function

- Production cost:

$$7400\ P_1 + 7500\ P_2 + 7600\ P_3 + 7800\ P_4$$

- Holding Cost:

$$120\ (I_1 + I_2 + I_3)$$

Constraints



$$P_t + I_{t-1} = d_t + I_t \quad \text{for } t = 1 \ldots T$$

Let $t = 1 \ldots T$ denote the production periods
Let P_t, I_t denote decision variables for production and inventory, respectively
Let c_t, d_t denote production costs and product demand, respectively

$$\min \sum_{t=1\ldots T} (c_t P_t + 120 I_t)$$

$$\begin{aligned}
\text{s.t.} \quad & P_t + I_{t-1} = d_t + I_t && \text{for } t = 1 \ldots T \\
& I_0 = 1000 \\
& I_T = 1500 \\
& 0 \leq P_t \leq 4000 && \text{for } t = 1 \ldots T \\
& I_t \geq 0 && \text{for } t = 1 \ldots T
\end{aligned}$$

```
# number of months in the planning horizon
param MONTHS;

# inventory holding cost per unit per month
param ic;

# cost of producing one ton in month i
param c {1 .. MONTHS};

# tons of product needed in month i
param d {1 .. MONTHS};
```

```
#DECISION VARIABLES

# tons produced in month i
# nonnegativity and max production limits
var P {1 .. MONTHS} >= 0, <= 4000;

# tons in inventory at the end of month I
# nonnegativity constraints
var I {0 .. MONTHS} >= 0;
```

```
#OBJECTIVE
#minimize production and inventory costs
minimize cost:
sum{i in 1..MONTHS} (c[i]*P[i] + ic*I[i]);

#CONSTRAINTS
#flow-balance constraint
subject to inventory {i in 1..MONTHS}:
   P[i] + I[i-1] = I[i] + d[i];

subject to initial_inventory: I[0] = 1000;

subject to final_inventory: I[MONTHS] = 1500;
```

```
data;

param MONTHS := 4;
param ic := 120;

param c :=
1 7400
2 7500
3 7600
4 7800;

param d :=
1 2400
2 2200
3 2700
4 2500  ;
```

optimal solution for NSC

Variable	Value
P1	2300
P2	4000
P3	4000
P4	0
I0	1000
I1	900
I2	2700
I3	4000
I4	1500

optimal objective

$78,332,000

multi period production planning

What if National Steel Corporation (NSC) had *multiple* products?

And wanted to schedule for 12 months?

Let K denote the set of products.

$$\min \sum_{t=1\ldots T, k\in K} (c_{kt}P_{kt} + 120I_{kt})$$

$$\begin{aligned}
\text{s.t.} \quad & P_{kt} + I_{k,t-1} = d_{kt} + I_{kt} && \text{for } t = 1\ldots T, k \in K \\
& I_{k,0} = B_k && \forall k \in K \\
& I_{kT} = F_k && \forall k \in K \\
& 0 \le P_{kt} \le 4000 && \text{for } t = 1\ldots T, k \in K \\
& I_{kt} \ge 0 && \text{for } t = 1\ldots T, k \in K
\end{aligned}$$

```
#set of products
set PRODUCTS;

#months in the planning horizon
param MONTHS;

#cost of producing one ton of product k in month i
param c {1 .. MONTHS, PRODUCTS};

#demand of product k in month i
param d {1 .. MONTHS, PRODUCTS};

var P {1 .. MONTHS, PRODUCTS} >= 0;
var I {0 .. MONTHS, PRODUCTS} >= 0;
```

```
minimize cost:
sum{i in 1..MONTHS, p in PRODUCTS}
         (c[i,p]*P[i,p] + 120*I[i,p]);

subject to inventory {i in 1 .. MONTHS, p in PRODUCTS}:
  P[i,p] + I[i-1,p] = d[i,p] + I[i,p];

subject to initial_inv {p in PRODUCTS}: I[0,p] = 1000;
subject to final_inv {p in PRODUCTS}:
      I[MONTHS,p] >= 1500;

subject to max_prod {i in 1 .. MONTHS, p in PRODUCTS}:
      P[i,p] <= 4000;
```

```
minimize cost:
sum{i in 1..MONTHS, p in PRODUCTS}
         (c[i,p]*P[i,p] + 120*I[i,p]);

subject to inventory {i in 1 .. MONTHS, p in PRODUCTS}:
  P[i,p] + I[i-1,p] = d[i,p] + I[i,p];

subject to initial_inv {p in PRODUCTS}: I[0,p] = 1000;
subject to final_inv {p in PRODUCTS}:
      I[MONTHS,p] >= 1500;

subject to max_prod {i in 1 .. MONTHS}:
      sum {p in PRODUCTS} P[i,p] <= 4000;
```

```
set PRODUCTS := steel al;
param MONTHS := 12;

param c:
        steel       al :=
1       7400        3400
2       7500        3500
3       7600        3600
4       7800        3800
5       7353        3199
6       7813        3015
7       7010        3747
8       7139        3445
9       7203        3932
10      7199        3466
11      7604        3419
12      7272        3846;
```

```
param d:
        steel       al :=
1       2400        1000
2       2200        1200
3       2700        1400
4       2500        1600
5       2762        1738
6       2456        1176
7       2019        1406
8       2821        1935
9       2445        1917
10      2615        1410
11      2792        1894
12      2922        1058;
```

Increase/Decrease Penalty

- **Suppose that if the production level is increased or decreased from one month to the next, then NSC incurs a cost for implementing these changes.**
- **Specifically, for each ton of increased or decreased production over the previous month, the cost is $50 (except for month 1).**

now with penalties…

Variable	Value
P1	2300
P2	4000
P3	4000
P4	0
I0	1000
I1	900
I2	2700
I3	4000
I4	1500

This solution would incur an extra cost (4000 – 2300) ($50) =$85,000 for increasing the production from 2300 to 4000 tons month 1 to month 2.

And (4000 – 0) ($50) =$200,000 for decreasing the production from 4000 to 0 tons month 3 to month 4.

new objective function

$$\min \quad 7400P_1 + 7500P_2 + 7600P_3 + 7800P_4 + 120\sum_{t=1}^{4} I_t +$$

$$50|P_1 - P_2| + 50|P_2 - P_3| + 50|P_3 - P_4|$$

?

To make the objective function linear define:

- Y_i = increase from month *i-1* to month *i*
- Z_i = decrease from month *i-1* to month *i*

$$\min 7400P_1 + 7500P_2 + 7600P_3 + 7800P_4 + 120\sum_{i=0}^{4} I_i$$

$$+ 50\sum_{i=2}^{4}\left(Y_i + Z_i\right)$$

Additional Constraints

$$Y_i \geq 0 \text{ for } i = 2, 3, 4$$

$$Z_i \geq 0 \text{ for } i = 2, 3, 4$$

$$Y_i - Z_i = P_i - P_{i-1} \text{ for } i = 2, 3, 4$$

Examples

1. If $P_1 = P_2$, then $Y_2 = 0$, and $Z_2 = 0$
2. If $P_1 = 2300$ and $P_2 = 4000$ then $Y_2 = 1700$, and $Z_2 = 0$
3. If $P_1 = 4000$ and $P_2 = 2300$ then $Y_2 = 0$, and $Z_2 = 1700$

Now, it is optimal to produce 2575 tons in each month and the total cost is $78,520,500.